# Type [331], CY of degree 17 via linkage -- lifting to a 3-fold with components of degrees 13 and 4

We construct via linkage a 3-fold $X = X_{13} \cup X_4 \subset \bf{P}^7$, which is an arithmetically Gorenstein Calabi-Yau 3-fold of degree 17 in $\PP^7$. For an artinian reduction $A_F$, the quadratic part of the ideal $F^\perp$ is the ideal of a plane and a point $p$, and $F^\perp$ contains a family of ideals of six points in the plane and the fixed point $p$. We construct $X_{13}$ in a web of cubics in a P6, and $X_{4}$ as a quartic in a P4. In the construction the intersection $X\cup X'$ of a component $X$ with the other is an anticanonical divisor on $X$.

The Betti table is

$\phantom{WWWW} \begin{matrix} &0&1&2&3&4\\\text{total:}&1&11&20&11&1\\\text{0:}&1&\text{.}&\text{.}&\text{.}&\text{.}\\ \text{1:}&\text{.}&3&3&1&\text{.}\\\text{2:}&\text{.}&7&14&7&\text{.}\\\text{3:}&\text{.}&1&3& 3&\text{.}\\\text{4:}&\text{.}&\text{.}&\text{.}&\text{.}&1\\ \end{matrix}$

This variety $X \subset \PP^7$ has two components, $X_{13}$ in a $\PP^6$ linked to a $\PP^3$ in the cubic pfaffians of a 7x7 skew matrix, and $X_4$, a quartic 3-fold in a $\PP^4$. $X_{13}$ and $X_4$ intersect in a hyperplane section of $X_4$.

 i1 : kk = QQ; i2 : U = kk[y0,y1,y2,y3,y4,y5,y6,y7]; i3 : u7 = ideal vars U; o3 : Ideal of U i4 : P6 = ideal(y0); -- a P6 o4 : Ideal of U i5 : P4 = ideal(y1,y2,y3); --a P4 o5 : Ideal of U

We now choose a skew symmetric $7 \times 7$ matrix of linear forms in $U$ whose cubic pfaffians contain a $\PP^4$.

 i6 : q1=(gens u7)*random(kk^8,kk^7); 1 7 o6 : Matrix U <--- U i7 : q2=(gens u7)*random(kk^8,kk^7); 1 7 o7 : Matrix U <--- U i8 : q3=(gens P4)*random(kk^3,kk^7); 1 7 o8 : Matrix U <--- U i9 : q4=(gens P4)*random(kk^3,kk^7); 1 7 o9 : Matrix U <--- U i10 : q5=(gens P4)*random(kk^3,kk^7); 1 7 o10 : Matrix U <--- U i11 : q6=(gens P4)*random(kk^3,kk^7); 1 7 o11 : Matrix U <--- U i12 : q7=(gens P4)*random(kk^3,kk^7); 1 7 o12 : Matrix U <--- U i13 : F1=transpose q1|transpose q2|transpose q3|transpose q4| transpose q5|transpose q6|transpose q7; 7 7 o13 : Matrix U <--- U i14 : F = F1-transpose F1;--a skew symmetric 7x7 matrix ; 7 7 o14 : Matrix U <--- U i15 : Y14 = pfaffians (6,F); --a 4-fold of degree 14 that contain P4 o15 : Ideal of U

The variety defined by the cubic pfaffians is a 4-fold, has degree 14, and contains the $\PP^4$.

 i16 : (dim Y14, degree Y14) o16 = (5, 14) o16 : Sequence i17 : isSubset(Y14, P4) o17 = true

We take the residual of the $\PP^4$ in $Y_{14}$. This is a 4-fold of degree 13.

 i18 : Y13 = Y14:P4;--a 4-fold of degree 13 o18 : Ideal of U i19 : dim Y13 o19 = 5 i20 : degree Y13 o20 = 13

We intersect this variety with the $\PP^4$, obtaining a quartic hypersurface $X_4$ in $\PP^4$.

 i21 : X4 = trim(Y13 + P4); o21 : Ideal of U

Let $X_{13}$ be the 3-fold which is the intersection of $Y_{13}$ with the hyperplane $y_0 = 0$.

 i22 : X13 = trim(P6 + Y13);-- A 3-fold of degree 13 in P6 o22 : Ideal of U i23 : dim X13 o23 = 4 i24 : degree X13 o24 = 13

The union $X$ of $X_{13}$ and $X_4$ is a 3-fold of degree 17 in $\PP^7$, with Betti table of type [331]

 i25 : X = intersect (X4,X13); --A 3-fold of degree 17 in P7, the union of X13 and X4, with betti table of type 331 o25 : Ideal of U i26 : dim X o26 = 4 i27 : degree X o27 = 17 i28 : betti res X 0 1 2 3 4 o28 = total: 1 11 20 11 1 0: 1 . . . . 1: . 3 3 1 . 2: . 7 14 7 . 3: . 1 3 3 . 4: . . . . 1 o28 : BettiTally

The intersection of $X_4$ and $X_{13}$ is a quartic surface in a $\PP^3$, which is a hyperplane section of $X_4$.

 i29 : Z = trim(X13 + X4); o29 : Ideal of U i30 : dim Z -- 3, therefore a surface in projective space o30 = 3 i31 : degree Z o31 = 4 i32 : dim X4 o32 = 4 i33 : X4 + ideal(y0) == Z o33 = true